
Preparations

We use ghc-7.4.1.

If you happen to have a really recent development snapshot of
ghc-7.5, that’s even better.

We don’t strictly need any Cabal packages, but

generic-deriving-1.2.1

is helpful.

Haven’t you ever wondered
how deriving works?



Equality on binary trees

data T = L | N T T

Let’s try ourselves:

eqT :: T→ T→ Bool
eqT L L = True
eqT (N x1 y1) (N x2 y2) = eqT x1 x2 && eqT y1 y2
eqT = False

Easy enough, let’s try another . . .

Equality on another type

data Choice = I Int | C Char | B Choice Bool | S Choice

eqChoice :: Choice→ Choice→ Bool
eqChoice (I n1 ) ( I n2 ) = eqInt n1 n2
eqChoice (C c1 ) (C c2 ) = eqChar c1 c2
eqChoice (B x1 b1) (B x2 b2) = eqChoice x1 x2 &&

eqBool b1 b2
eqChoice (S x1 ) (S x2 ) = eqChoice x1 x2
eqChoice = False

Do you see a pattern?

A pattern for defining equality

I How many cases does the function definition have?
I What is on the right hand sides?

I How many clauses are there in the conjunctions on each
right hand side?

Relevant concepts:

I number of constructors in datatype,
I number of fields per constructor,
I recursion leads to recursion,
I other types lead to invocation of equality on those types.

More datatypes

data Tree a = Leaf a | Node (Tree a) (Tree a)

Like before, but with labels in the leaves.

How to define equality now?

We need equality on a !

eqTree :: (a→ a→ Bool)→ Tree a→ Tree a→ Bool
eqTree eqa (Leaf n1 ) (Leaf n2 ) = eqa n1 n2
eqTree eqa (Node x1 y1) (Node x2 y2) = eqTree eqa x1 x2 &&

eqTree eqa y1 y2
eqTree eqa = False

Type classes

Note how the definition of eqTree is perfectly suited for a type
class instance:

instance Eq a⇒ Eq (Tree a) where
(= =) = eqTree (= =)

In fact, type classes are usually implemented as dictionaries,
and an instance declaration is translated into a dictionary
transformer.

.

Yet another equality function

This is often called a rose tree:

data Rose a = Fork a [Rose a]

Let’s assume we already have:

eqList :: (a→ a→ Bool)→ [a]→ [a]→ Bool

How to define eqRose ?

eqRose :: (a→ a→ Bool)→ Rose a→ Rose a→ Bool
eqRose eqa (Fork x1 xs1) (Fork x2 xs2) =

eqa x1 x2 && eqList (eqRose eqa) xs1 xs2

No fallback case needed because there is only one constructor.

More concepts

I Parameterization of types is reflected by parameterization
of the functions.

I Application of parameterized types is reflected by
application of the functions.

.

.Well-Typed



The equality pattern
An informal description

In order to define equality for a datatype:

I introduce a parameter for each parameter of the datatype,
I introduce a case for each constructor of the datatype,
I introduce a final catch-all case returning False ,
I for each of the other cases, compare the constructor fields

pair-wise and combine them using (&&) ,
I for each field, use the appropriate equality function;

combine equality functions and use the parameter
functions as needed.

If we can describe it, can we write a program to do it?

.

Interlude:
type isomorphisms



Isomorphism between types

Two types A and B are called isomorphic if we have functions

f :: A→ B
g :: B→ A

that are mutual inverses, i.e., if

f ◦ g ≡ id
g ◦ f ≡ id

.

Example
Lists and Snoc-lists are isomorphic

data SnocList a = Lin | SnocList a :> a

listToSnocList :: [a]→ SnocList a
listToSnocList [ ] = Lin
listToSnocList (x : xs) = listToSnocList xs :> x
snocListToList :: SnocList a→ [a]
snocListToList Lin = [ ]
snocListToList (xs :> x ) = x : snocListToList xs

We can prove that these are inverses.

.

The idea of datatype-generic programming

I Represent a type A as an isomorphic type Rep A .

I If a limited number of type constructors is used to build
Rep A ,

I then functions defined on each of these type constructors
I can be lifted to work on the original type A
I and thus on any representable type.

.

The idea of datatype-generic programming – contd.

In fact, we do not even quite need an isomorphic type.

For a type A , we need a type Rep A and from :: A→ Rep A

and to :: Rep A→ A such that

to ◦ from ≡ id

We call such a combination an embedding-projection pair.

.

Choice between constructors

Which type best encodes choice between constructors?

Well, let’s restrict to two constructors first.

Booleans encode choice, but do not provide information what
the choice is about.

data Either a b = Left a | Right a

Choice between three things:

type Either3 a b c = Either a (Either b c)

.

Combining constructor fields

Which type best encodes combining fields?

Again, let’s just consider two of them.

data (a, b) = (a, b)

Combining three fields:

type Triple a b c = (a, (b, c))

What about constructors without arguments?

We need another type.

Well, how many values does a constructor without argument
encode?

data () = ()

Representing types



Representing types

To keep representation and original types apart, let’s define
isomorphic copies of the types we need:

data U = U
data a :+: b = L a | R b
data a :∗ : b = a :∗ : b

We can now get started:

data Bool = False | True

How do we represent Bool ?

type RepBool = U :+: U

.

A class for representable types

class Generic a where
type Rep a
from :: a→ Rep a
to :: Rep a→ a

The type Rep is an associated type. GHC allows us to define
datatypes and type synonyms within classes, depending on the
class parameter(s).

This is equivalent to defining Rep separately as a type family:

type family Rep a

.

Representable Booleans

instance Generic Bool where
type Rep Bool = U :+: U
from False = L U
from True = R U
to (L U) = False
to (R U) = True

Question

Are Bool and Rep Bool isomorphic?

.

.Well-Typed



Representable Booleans

instance Generic Bool where
type Rep Bool = U :+: U
from False = L U
from True = R U
to (L U) = False
to (R U) = True

Question

Are Bool and Rep Bool isomorphic?

.

.Well-Typed



Representable lists

instance Generic [a] where
type Rep [a] = U :+: (a :∗ : [a])
from [ ] = L U
from (x : xs) = R (x :∗ : xs)
to (L U ) = [ ]
to (R (x :∗ : xs)) = x : xs

Note that the representation of recursive types mentions the
original types – if needed, we can apply the transformation
multiple times.

Note further that we do not require Generic a .

Representable trees

instance Generic (Tree a) where
type Rep (Tree a) = a :+: (Tree a :∗ : Tree a)
from (Leaf n ) = L n
from (Node x y ) = R (x :∗ : y)
to (L n ) = Leaf n
to (R (x :∗ : y)) = Node x y

.

Representable rose trees

instance Generic (Rose a) where
type Rep (Rose a) = a :∗ : [Rose a]
from (Fork x xs) = x :∗ : xs
to (x :∗ : xs ) = Fork x xs

.

Representing primitive types

For some types, it does not make sense to define a structural
representation – for such types, we will have to define generic
functions directly.

instance Generic Int where
type Rep Int = Int
from = id
to = id

.

Back to equality



Intermediate summary

I We have defined class Generic that maps datatypes to
representations built up from U , (:+:) , (:∗ :) and other
datatypes.

I If we can define equality on the representation types, then
we should be able to obtain a generic equality function.

I Let us apply the informal recipe from earlier.

.

Equality on sums

eqSum :: ( a → a → Bool)→
( b→ b→ Bool)→

a :+: b→ a :+: b→ Bool
eqSum eqa eqb (L a1) (L a2) = eqa a1 a2
eqSum eqa eqb (R b1) (R b2) = eqb b1 b2
eqSum eqa eqb = False

Equality on products

eqProd :: ( a → a → Bool)→
( b→ b→ Bool)→

a :∗ : b→ a :∗ : b→ Bool
eqProd eqa eqb (a1 :∗ : b1) (a2 :∗ : b2) =

eqa a1 a2 && eqb b1 b2

.

Equality on units

eqUnit :: U→ U→ Bool
eqUnit U U = True

.

What now?



A class for generic equality

class GEq a where
geq :: a→ a→ Bool

instance (GEq a, GEq b)⇒ GEq (a :+: b) where
geq = eqSum geq geq

instance (GEq a, GEq b)⇒ GEq (a :∗ : b) where
geq = eqProd geq geq

instance GEq U where
geq = eqUnit

Instances for primitive types:

instance GEq Int where
geq = eqInt

.

Dispatching to the representation type

eq :: (Generic a, GEq (Rep a))⇒ a→ a→ Bool
eq x y = geq (from x) (from y)

.

Dispatching to the representation type

eq :: (Generic a, GEq (Rep a))⇒ a→ a→ Bool
eq x y = geq (from x) (from y)

Defining generic instances is now trivial:

instance GEq Bool where
geq = eq

instance GEq a⇒ GEq [a] where
geq = eq

instance GEq a⇒ GEq (Tree a) where
geq = eq

instance GEq a⇒ GEq (Rose a) where

Dispatching to the representation type

eq :: (Generic a, GEq (Rep a))⇒ a→ a→ Bool
eq x y = geq (from x) (from y)

Or with the DefaultSignatures language extension:

class GEq a where
geq :: a→ a→ Bool
default geq :: (Generic a, GEq (Rep a))⇒ a→ a→ Bool
geq = eq

instance GEq Bool
instance GEq a⇒ GEq [a]
instance GEq a⇒ GEq (Tree a)
instance GEq a⇒ GEq (Rose a)

Have we won
or

have we lost?



Amount of work

Question

Haven’t we just replaced some tedious work (defining equality
for a type) by some other tedious work (defining a
representation for a type)?

Yes, but:

I The representation has to be given only once, and works
for potentially many generic functions.

I Since there is a single representation per type, it could be
generated automatically by some other means (compiler
support, TH).

I In other words, it’s sufficient if we can use deriving on

class Generic .

.

So can we derive Generic ?

Yes! (With DeriveGeneric.)

But it’s not quite as simple as we’ve seen before:

.

So can we derive Generic ?

Yes! (With DeriveGeneric.)

But it’s not quite as simple as we’ve seen before:

class Generic a where
type Rep a :: ∗ → ∗
from :: a→ Rep a x
to :: Rep a x→ a

Representation types are now of kind ∗ → ∗ .

An extra argument?

Having an extra argument is an admittedly somewhat
questionable, but pragmatic choice:

I We are not only interested in deriving classes
parameterized by types, but also classes by type
constructors.

I Haskell’s kind system doesn’t (well, didn’t) support kind
polymorphism.

I The current choice allows two representations, one for fully
applied types of kind ∗ , one for type constructors of
kind ∗ → ∗ , to share a single set of representation type
constructors.

For the beginning, we can just try to “ignore” the additional
argument.

.

.Well-Typed



An extra argument?

Having an extra argument is an admittedly somewhat
questionable, but pragmatic choice:

I We are not only interested in deriving classes
parameterized by types, but also classes by type
constructors.

I Haskell’s kind system doesn’t (well, didn’t) support kind
polymorphism.

I The current choice allows two representations, one for fully
applied types of kind ∗ , one for type constructors of
kind ∗ → ∗ , to share a single set of representation type
constructors.

For the beginning, we can just try to “ignore” the additional
argument.

.

.Well-Typed



Simple vs. GHC representation

Old:

type instance Rep (Tree a) = a :+: (Tree a :∗ : Tree a)

New:

type instance Rep (Tree a) =
M1 D D1Tree
(M1 C C1_0Tree

(M1 S NoSelector (K1 P a))
:+:
M1 C C1_1Tree
(M1 S NoSelector (K1 R (Tree a))
:∗ :
M1 S NoSelector (K1 R (Tree a))
)

)
.

Familiar components

Everything is now lifted to kind ∗ → ∗ :

data U1 a = U1
data (f :+: g) a = L1 (f a) | R1 (g a)
data (f :∗ : g) a = f a :∗ : g a

.

Wrapping constant types

This is an extra type constructor wrapping every constant type:

newtype K1 t c a = K1 {unK1 :: c}
data P -- marks parameters
data R -- marks other occurrences

The first argument t is not used on the right hand side. It is
supposed to be instantiated with either P or R .

.

Meta information

newtype M1 t i f a = M1 {unM1 :: f a}
data D -- marks datatypes
data C -- marks constructors
data S -- marks (record) selectors

Depending on the tag t , the position i is to be filled with a
datatype belonging to class Datatype , Constructor , or

Selector .

.

Meta information – contd.

class Datatype d where
datatypeName :: w d f a→ String
moduleName :: w d f a→ String

instance Datatype D1Tree where
datatypeName = "Tree"
moduleName = ...

Similarly for constructors.

.

Adapting the equality class(es)

Works on representation types:

class GEq′ f where
geq′ :: f a→ f a→ Bool

Works on “normal” types:

class GEq a where
geq :: a→ a→ Bool
default geq :: (Generic a, GEq′ (Rep a))⇒ a→ a→ Bool
geq x y = geq′ (from x) (from y)

Instance for GEq Int and other primitive types as before.

.

Adapting the equality class(es) – contd.

instance (GEq′ f, GEq′ g)⇒ GEq′ (f :+: g) where
geq′ (L1 x) (L1 y) = geq′ x y
geq′ (R1 x) (R1 y) = geq′ x y
geq′ = False

Similarly for :∗ : and U1 .

An instance for constant types:

instance GEq a⇒ GEq′ (K1 t a) where
geq′ (K1 x) (K1 y) = geq x y

.

Adapting the equality classes – contd.

For equality, we ignore all meta information:

instance GEq′ f⇒ GEq′ (M1 t i f) where
geq′ (M1 x) (M1 y) = geq′ x y

All meta information is grouped under a single datatype, so that
we can easily ignore it all if we want to.

Functions such as show and read can be implemented
generically by accessing meta information.

.

Example functions

Many example functions are defined in the package
generic-deriving.

Many related representations are used in other packages on
Hackage, such as instant-generics or regular or
multirec.

For these, the representations cannot be generated
automatically by GHC (but usually by Template Haskell).

.

Constructor classes

To cover classes such as Functor , Traversable , Foldable
generically, we need a way to map between a type constructor
and its representation:

class Generic1 f where
type Rep1 f :: ∗ → ∗
from1 :: f a→ Rep1 f a
to1 :: Rep1 f a→ f a

Use the same representation type constructors, plus

data Par1 p = Par1 {unPar1 :: p }
data Rec1 f p = Rec1 {unRec1 :: f p}

GHC from version 7.6 will be able to derive Generic1 , too.

.

